
Поняття сортування
Сортуванням взагалі називають процес впорядковування множини елементів за певною ознакою. Наприклад, у житті ми стикаємося з таким процесом, коли хочемо відсортувати фрукти за якістю на вітрині, учнів за зростом на уроці фізкультури, банки із фарбами за їх кольором при ремонті тощо. В математиці при побудові многокутника за даними у довільному порядку координатами вершин їх зручно попередньо відсортувати за певною ознакою, наприклад за зростанням абсцис, спаданням ординат, відстанню до початку координат, кутом нахилу радіус-вектора вершини до осі чи будь-якою іншою ознакою.

Саму ж ознаку, за якою виконується сортування, називають ключем сортування. Так, при впорядкуванні слів у словнику ключем є їх  лексикографічний порядок (ми кажемо, що слова розміщуються у алфавітному порядку); список учнів може бути впорядкований за їхніми прізвищами по алфавіту, роком народження (від найстарших до наймолодших), домашньою адресою або будь-якою іншою ознакою. Дати зазвичай впорядковуються за таким ключем: РРРРММДД, де РРРР – рік (чотиризначне число), ММ – місяць (двозначне число), ДД – день (двозначне число).

Як бачимо, об’єктами впорядкування можуть бути найрізноманітніші елементи. Головним є визначення відношення порядку на сортованій множині елементів, тобто для будь-яких двох елементів множини А і В необхідно виділити закон визначення їх взаємного правильного розміщення (об’єкт A має стояти після об’єкту B чи перед ним).

Як відомо, всі дані, що зберігаються на комп’ютері, оцифровані, тобто представлені у числовому вигляді, тому традиційно програмісти, говорячи про сортування, розуміють впорядкування за зростанням або спаданням числових величин.

Для впорядкування множини елементів розроблено багато алгоритмів сортування, які мають велике практиче значення. Завдяки впорядкуванню значно спрощується пошук необхідних даних у великих інформаційних системах, з’являються спрощені методи їх обробки. 

У навчальному процесі ця тема теж має дуже велике значення, оскільки учням, що опановують премудрощами мови програмування, надається можливість реалізувати вже відомі алгоритми, використовуючи мову, що вивчається, як інструмент. Фактично у шкільній програмі ця тема вивчається тоді, коли учні вже знають основні структури даних (скалярні величини та структурований тип даних масив) і базові конструкції мови програмування (циклічні, розгалужені та допоміжні алгоритми). Тому перед ними постає питання, як відомі з життя дії описати, використовуючи особливості комп’ютерного „мислення” та мови програмування. Учням можна дати самостійно пофантазувати і спробувати придумати свій власний алгоритм впорядкування даних. І хоча, як правило, придуманий метод впорядкування буде описувати відомий з життя алгоритм, це дуже заохочує учнів до роботи, зацікавлює їх складним мистецтвом створення нових алгоритмів та їх реалізації.
Традиційно розрізняють внутрішнє та зовнішнє сортування. Перший різновид впорядковує дані, що знаходяться в оперативній (внутрішній) пам’яті комп’ютера, а другий оперує із даними на зовнішніх носіях такого об’єму, для якого не вистачає оперативної пам’яті.

Почнемо розгляд методів впорядкування із внутрішніх методів сортування. Всі ці методи можна класифікувати за кількома ознаками. Так, існують методи, що виконують сортування даних у тому ж масиві, в якому їх отримали (сортування на місці), але є і такі, що заповнюють інший масив за деяким правилом, не порушуючи початковий набір даних. Очевидно, що перший з методів оптимальніший за використанням ресурсів пам’яті, проте, як побачимо далі, сам алгоритм буде або складнішім, або довшим за часом виконання.

Усі алгоритми сортування також поділяються на базові – ті, що працюють порівняно повільно, проте не потребують великих зусиль на реалізацію і відлагодження – та швидкі – ті, що сортують великі масиви даних досить швидко, але зазвичай мають достатньо складний у розумінні та громіздкий алгоритм.

Зверніть також увагу на те, що зберігатися дані в оперативній пам’яті можуть по різному. Відомо, що у більшості мов програмування можливе як статичне, так і динамічне виділення пам’яті. У першому випадку оперативна пам’ять для зберігання масиву даних виділяється на початку роботи програми і вважається зайнятою під час її роботи. Цей метод виділення пам’яті простіший, але не економічний, оскільки на початку роботи програми необхідно зарезервувати стільки пам’яті, скільки вистачить на будь-який об’єм даних у деяких межах. У другому випадку пам’ять для збереження даних виділяється за необхідністю (динамічно), тому цей метод є раціональнішим хоча і складнішим.
Майже всі алгоритми внутрішього сортування можна реалізувати на обох типах наборів даних. Далі, при описі методів сортування ми будемо виділяти деякі відмінності в алгоритмах для тих чи інших структур даних.

Ще один нюанс: очевидно, що всі алгоритми сортування “на місці” (а такими є перевежна більшість із відомих) потребують взаємного обміну двох елементів. Задача тривіальна, але вже тут можна пофантазувати і придумати кілька методів виконання цієї дії.

Стандартний метод – це використання третьої змінної. Елементи масиву даних мають однаковий тип (адже нереально сортувати цукерки та зошити за одією тією ж самою ознакою), тому й змінна, що використовується для обміну, теж повинна мати той самий тип. Тоді допоміжний алгоритм матиме, очевидно, наступний вигляд:

Procedure Swap (var a,b:integer);

Var t:integer;

Begin

  t:=a: a:=b; b:=t;

End;
Цей метод найбільш розповсюджений серед програмістів, він є універсальним і взагалі не викликає ніяких проблем. В подальшому саме його ми будемо використовувати як вже відомий при складанні алгоритмів сортування.
Тільки з точки зору розвитку фантазії та відриву від стереотипів пропонуємо інший метод. Він випливає з такої життєвої ситуації: наприклад, на дуже маленькому столі лежать дві книжки (вільного місця вже нема), а вам необхідно однією рукою поміняти їх місцями. Що можна зробити в такому випадку? Кмітливий зрозуміє, що треба покласти одну книжку зверху іншої, потім пересунути цю стопку на вільне місце, а потім покласти верхню книжку на місце, яке звільнилося в результаті пересування. Як використати цю ситуацію на числах? Подивимось на конкретному прикладі:

	Перше число (а)
	Друге число (b)
	Виконувана дія
	Примітки

	12
	45
	–
	Вихідне положення

	12 + 45 = 57
	45
	a := a+b
	“Книжки” покладені одна на одну

	57
	57 – 45 = 12
	b := a–b
	Стопка пересунута на вільне місце і верхня “книжка” знята

	57 – 12 = 45
	12
	a := a–b
	Кінцеве положення: “книжки” помінялися місцями


Пропонуємо написати програму на запропонований алгоритм самостійно.

Примітка: Майже аналогічно до операції додавання, але з деякими обмеженнями,  можна використовувати операції віднімання, множення, or, xor чи and. За бажанням, для власного самовдосконалення, можете спробувати реалізувати їх теж.
Аналіз алгоритмів
Іноді для вибору найефективнішого з кількох алгоритмів розв’язання тієї самої задачі корисно проаналізувати, скільки пам’яті або часу вони потребують для свого виконання. Очевидно, що час виконання алгоритмів сортування залежить від кількості елементів у вхідному масиві та ступені його впорядкованості (швидкість процесора переважно не враховують, оскільки невідомо, на якому комп’ютері буде виконуватись програма). Ступінь впорядкованості вхідного масиву передбачити не можна аж ніяк (вона може бути будь-якою), тому треба завжди розраховувати на найгірший випадок.

Складністю алгоритму називається порядок числової залежності кількості операцій, виконаних під час роботи програми, від яких-небудь вхідних параметрів. При аналізі алгоритмів сортування таким параметром є кількість елементів у масиві N. Складність алгоритму впорядкування позначатимемо Θ (f(N)), де f(N) – певна функція, що характеризує кількість операцій, виконаних під час дії алгоритму.
Базові методи сортування
Різні алгоритми базових методів сортувань можна класифікувати за принципом їх виконання:
1) обмінні сортування – виконується перегляд усього масиву, під час якого порівнюються пари елементів, причому не обов’язково розташованих поруч, і, якщо це сприяє впорядкуванню, виконується їх обмін;

2) сортування вибором – з масиву даних вибирається елемент із заданою ознакою (наприклад, найменший чи найбільший з усіх), після чого він ставиться на своє правильне місце, тобто місце, на якому стоятиме у впорядкованому масиві. Після цього попередній крок повторюється для частини масиву, яка залишилася невпорядкованою;

3) сортування вставкою – елементи з набору даних розглядаються підряд, але одразу визначається місце у впорядкованій частині масиву, на якому повинен стояти конкретно взятий елемент, після чого саме на це місце він і встановлюється.

Кожен із запропонованих різновидів сортування можна реалізувати великою кількістю методів, більш чи менш ефективних в залежності від вхідних даних. Варто запропонувати учням спробувати свої сили у реалізації більшості із запропонованих, оскільки це розвиває їх алгоритмічне мислення та підвищує техніку програмування в цілому.
Примітка: Далі при описі методів сортування ми вважатимемо, що перед нами стоїть задача відсортувати лінійний масив, що містить N_max цілих чисел, за зростанням. У процедуру сортування буде передаватися масив вхідних даних (var A:IntArray) та кількість елементів, що підлягають сортуванню (N:integer), де IntArray – це користувацький тип даних, що описується у розділі опису типів, N_max – максимально можлива кількість елементів масиву (описується в розділі опису констант для статичних масивів):
Const N_max=1000;
Type

  IntArray = array [1..N_Max] of integer;
Обмінне сортування
Як правило вважається, що найпростішим у розумінні є алгоритм обмінного сортування. Він полягає у тому, що здійснюється прохід по вхідних даних, під час якого порівнюються два елементи. Якщо вони стоять “не правильно”, тобто більший за значенням знаходиться лівіше меншого (при сортуванні за зростанням), то здійснюється їх взаємний обмін. Очевидно, що в загальному випадку за один прохід по масиву досягти повного впорядкування не вдається, тому процес повторюється кілька разів.

За базовим алгоритмом, при сортуванні статичного набору даних (масиву) можна, виконавши N-1 проходів по масиву, досягти успіху, оскільки при цьому кожен елемент масиву буде порівняний з усіма іншими і стоятиме, відповідно, на правильному місці. Реалізація алгоритму матиме наступний вигляд:

Procedure BubbleSort (var A:IntArray;N:integer);

Var i,j:integer;

Begin

  For i:=1 to N-1 do

  For j:=1 to N-1 do

    If A[j]>A[j+1] 
    then Swap (A[j],A[j+1]);

End;

Та очевидно, що цей алгоритм можна вдосконалити. Для цього розглянемо, як переміщуються елементи по масиву під час виконання сортування:

	Початко-вий масив
	
	Прохід 1
	
	Прохід 2
	
	Прохід 3
	
	Прохід 4
	
	Прохід 5
	
	Прохід 6
	
	Прохід 7
	
	Прохід 8

	703

765

677

612

509

154

426

653

275

897

170

908

061

512

087

503
	
	908

426

653

275

512

061

503

087
	
	908

061

087

677

154

170

087

061
	
	908

897

765

703

677

653

612

512

509

503

426

275

170

154

087

061


Зверніть увагу на те, що не обов’язково кожен раз доходити до кінця масиву, адже, як видно на малюнку, при кожному з проходів в кінець масиву “спливає” найбільший елемент (не дарма цей метод ще називають методом “бульбашки”). Отож, внутрішній цикл можна примусити закінчуватися раніше, тобто праву границю циклу зменшувати на кожному кроці щонайменше на 1:

Procedure BubbleSort (var A:IntArray;N:integer);
Begin

  For i:=1 to N-1 do

  For j:=1 to N-i do

    If A[j]>A[j+1] 
    then Swap (A[j],A[j+1]);

End;

Але це ще не все. В деяких випадках може статися така ситуація, що спливаючий елемент потрапить на своє місце задовго до кінця масиву і після цього обміни вже не здійснюються (наприклад, дивись на малюнку прохід 4). Коли таке відбувається? Коли права частина масиву (в нашому випадку перегляду зліва направо) містить уже впорядкований набір даних. Тоді очевидним вдосконаленням методу буде запам’ятовування індекса останнього елемента, що підлягав перестановці, з наступним використанням його у якості правої межі невідсортованої частини масиву. Алгоритм у цьому випадку прийме наступний вигляд:

Procedure BubbleSort (var A:IntArray;N:integer);
Var i,j,Lim,Help:integer;

Begin

  Lim:=N-1;
  For i:=1 to N-1 do
  Begin
   For j:=1 to Lim do

    If A[j]>A[j+1] 
    then begin

          Swap (A[j],A[j+1]);

          Help:=j;

         end;
  Lim:=Help;
  End;

End;

У запропонованому фрагменті програми змінна Help використовується для тимчасового запам’ятовування індекса останнього праворуч елемента, що підлягав перестановці. Але оскільки цей індекс може змінюватись кілька разів (подивіться на приклад), а нам необхідно знати тільки останній праворуч елемент, що підлягав перестановці, ми не можемо використовувати змінну Help у якості параметру закінчення циклу. Тому у процедурі введена ще одна локальна змінна Lim, яка і організовує вихід з циклу For.
Із внутрішнім циклом, напевне, більше нічого не зробиш для його вдосконалення. Спробуємо тепер оптимізувати зовнішній цикл.

Розглянемо такий набір даних:

12   24   34   56   100   120   500
Одразу ж видно, що він вже відсортований за зростанням, і тому виникає питання: навіщо сортувати вже впорядкований набір? Але як визначити, що масив не потребує впорядкування? Адже порівняти всі елементи одночасно не можливо і тому зробити висновок про впорядкованість масиву теж не можна. 

Висновок про те, що масив вже впорядкований можна зробити у тому випадку, коли при черговому проході по масиву не зроблено жодного обміну (тобто всі елементи масиву впорядковані один відносно одного). Як же тепер зафіксувати момент досягнення впорядкованості? Для цього в програмуванні використовується таке поняття як прапорець.

Прапорцем називають змінну, яка може приймати два значення: 0 або 1 (істина true чи хибність false). Їй надають одне зі значень, якщо подія відбулась, і протилежне значення, якщо вона не відбулась. Наприклад, домовимось, що змінна Sorted буде набувати значення false, якщо виконана хоч одна перестановка, і true,  у протилежному випадку. Тоді алгоритм зміниться в такий спосіб:

Procedure BubbleSort (var A:IntArray;N:integer);
Var j,Lim,Help:integer;

    Sorted:boolean;

Begin

 Lim:=N-1;

 Repeat

  Sorted:=true;
  For j:=1 to Lim do

    If A[j]>A[j+1] 
    then begin

          Swap (A[j],A[j+1]);

          Help:=j;

          sorted:=false;

         end;
  Lim:=Help;
 Until sorted;

End;
Ще один різновид вдосконалення базового алгоритму – так званий метод “шейкера”. Він має особливий виграш, коли багато елементів вхідного масиву даних стоять далеко від правильних своїх позицій. Припустимо, нам необхідно відсортувати за зростанням такий набір:

100  60  25  5  10  5  2  3

При алгоритмі сортування, описаному вище, очевидно, що великі числа (100, 60) стануть на свої місця за два проходи („спливуть” наче бульбашки). Проте маленькі (2, 3) будуть просуватися на свої місця повільно: на одну позицію ліворуч за прохід, оскільки прохід по масиву здійснюється зліва направо Щоб прискорити просування маленьких чисел справа наліво до початку масиву, оптимальним буде організувати ще однин прохід – з останнього елемента до першого. Цей прохід дозволить мінімальному елементу швидко стати на своє  місце. Крім цього можна використати той самий прийом, що й раніше для зменшення кількості проходів внутрішнього циклу: запам’ятовувати номер останнього елементу, що переставлявся, для використання його індексу у якості границі невідсортованої частини. Масив буде відсортований не більше, ніж за N/2 таких “подвійних” проходів, але використовуючи прапорець, можна ще покращити алгоритм, закінчуючи впорядкування, коли перестановки вже не здійснюються. Алгоритм зміниться тоді до наступного вигляду:

Procedure ShakerSort (var A:IntArray;N:integer);
Var j,LimL,LimR,Help:integer;

    Sorted:boolean;

Begin

 LimL:=1; LimR:=N-1;

 Repeat

  sorted:=true;
  For j:=LimL to LimR do

   If A[j]>A[j+1] 
   then begin

         Swap (A[j],A[j+1]);

         Help:=j-1;

         Sorted:=false;

        end;

  LimR:=Help;
  If not sorted then

    For j:=LimR downto LimL do

      If A[j]>A[j+1] 
      then begin

            Swap (A[j],A[j+1]);

            Help:=j+1;

            Sorted:=false;

           end;

    LimL:=Help;
 Until sorted;

End;
Текст програми, як бачимо, став значно більшим, але швидкість сортування помітно зростає, якщо на вхід подано достатньо великий об’єм даних.
Примітка: При розгляді запропонованих алгоритмів учні повинні чітко усвідомити, що наявність двох циклів для сортування одновимірного масиву обгрунтована необхідністю організувати кілька (зовнішній цикл) проходів по масиву (внутрішній цикл). Кожен з них не залежить від іншого і тому запропоновані оптимізації можна використовувати як разом, так і окремо. Наприклад, оптимізація з використанням прапорця може не передбачати зменшення кількості проходів внутрішнього циклу, тобто внутрішній цикл може виконуватись N-1 разів, але при черговому проході без перестановок елементів прапорець буде набувати значення false і завершувати виконання алгоритму. Аналогічно, можна взагалі не брати додаткову змінну прапорець, якщо запам’ятовувати місце останньої перестановки елементів масиву, а безпосередньо закінчувати роботу у випадку, коли ця змінна набуває певного значення. Пропонуємо Вам самостійно придумати, як реалізувати в цьому випадку алгоритм.
Сортування вибором
При сортуванні вибором за зростанням серед елементів невпорядкованої частини масиву ми знаходимо максимум, який стає на своє місце у кінці цієї частини. При реалізації можна користуватися двома різними методами встановлення елемента на своє місце.

Перший. Припустимо, що у кінці невідсортованої частини масиву вже стоїть максимум. Проходимо по масиву з початку і до кінця і, якщо знаходимо елемент, більший за останній, обмінюємо їх між собою. Таким чином, після чергового проходу ми однозначно помістимо максимум в кінець. Далі попередня операція повторюється для частини масиву, що залишилась.

Реалізація описаного алгоритму доволі прозора:

Procedure SelectSort (var A:IntArray;N:integer);
Var i,j:integer;

Begin
  For i:=N downto 2 do
  For j:=1 to i-1 do
    If A[j]>A[i] 
    then Swap (A[i],A[j]);

End;

Недоліком цього способу є багаторазове використання операції Swap, адже, якщо порахувати, у найгіршому випадку обмін відбувається стільки ж разів, скільки у неоптимізованій “бульбашці”, тобто складність алгоритму становить N2 операцій.
Другий. Цей метод принципово відрізняється тим, що ми не переносимо перший знайдений елемент, більшій за останній, у кінець, а лише запам’ятовуємо в окремій змінній LimI його індекс. Сам елемент на місце ми ставимо вже після закінчення проходу по всьому масиву процедурою Swap, а тому обмін місцями двох змінних відбувається тільки коли це дійсно необхідно.
Реалізація зрозуміла:

Procedure SelectSort (var A:IntArray;N:integer);
Var i,j,LimI:integer;

Begin
  For i:=N downto 2 do

   Begin

    LimI:=1;
    For j:=1 to i do
     If A[j]>A[LimI] then LimI:=j;

    Swap (A[LimI],A[i]);

   End;

End;

Аналогічно модифікації для “бульбашки” – методу “шейкера”, можна одночасно шукати два граничні елементи, щоб не витрачати час на зайві проходи. Якщо у двох змінних LimI1 та LimI2 зберігати індекси найбільшого та найменшого елемента невідсортованої частини даних, а потім “розкидати” ці два елементи на свої місця в кінцях масиву, то за один прохід ми скорочуємо межі сортування на 2, а це вже суттєвий виграш.

Очевидні зміни алгоритму:

Procedure SelectSort (var A:IntArray;N:integer);
Var i,j,LimI1,LimI2:integer;

Begin
  For i:=1 to N div 2 do

  Begin

    LimI1:=1; LimI2:=1;
    For j:=i to N-i+1 do

    Begin
     If A[j]>A[LimI1] then LimI1:=j;

     If A[j]<A[LimI2] then LimI2:=j;

    End;

    Swap (A[N-i+1],A[LimI1]);

    Swap (A[i],A[LimI2]);

  End;

End;

Сортування вставками
Цей метод впорядкування елементів відрізняється від попереднього тим, що в наборі даних розшукується не “елемент для місця”, а “місце для елемента”. Тобто для кожного елемента знаходиться за певним алгоритмом місце його розташування у впорядкованому наборі. Яким же чином можна визначити місце елемента у відсортованому масиві, якщо набір даних не впорядкований? Очевидно, що можна для кожного з елементів підрахувати кількість значень масиву, що не перевищують даний, і тоді  місце знаходження елементу визначається однозначно. Так, у наступному наборі даних
2

для першого елемента (2) існує три значення, що менші за 2 (-3, 0, -13), значить у впорядкованому наборі даних двійка буде стояти на четвертому місці. Можна записати цей елемент у новий масив, після чого продовжити підрахунок кількості елементів, що не перевищують значення наступного елемента (-3) і так далі.

Реалізація запропонованого алгоритму не викликає труднощів, але у випадку, коли деякі елементи повторюються, наприклад, у такому наборі 

може виникнути неоднозначність для другого та п’ятого елементів масиву, значення яких однакові (2). Цей випадок можна обробити по-різному (спробуйте самостійно знайти вихід), але ми пропонуємо тільки один з них. Для реалізації цього методу нам будуть потрібні дві змінні count_eq, що підраховує кількість елементів, рівних даному, та count_less, що містить кількість елементів, менших за даний. Тоді алгоритм стає очевидним:
Procedure CountSort (var A:IntArray;N:integer;var B:IntArray);
Var i,j,count_eq,count_less:integer;

Begin
  For i:=1 to N do

  Begin

    Count_eq:=0; count_less:=0;

    For j:=1 to N do

    Begin
     If A[i]>A[j] then inc(count_less);

     If A[i]=A[j] then inc(count_eq);
    End;

    For j:=1 to count_eq do B[j+count_less]:=A[i];
  End;

End;

Запропонований алгоритм є прозорим для розуміння, аде є дуже неоптимальним, якщо масив містить багато елементів, що повторюються. Адже для кожного з повторюваних елементів у результуючий масив значення буде заноситись стильки разів, скільки разів повторюється елемент. З цієї ситуації можна знайти вихід, але це дещо ускладнить алгоритм. Наприклад, можна підраховувати кількість тільки тих елементів рівних розглядуваному, що знаходяться до нього. Нескладне удосконалення Ви можете зробити самостійно, але навіть в цьому випадку залишається інший недолік: це використання додаткового масиву. Тому алгоритм має цінність тільки як навчальний, і у реальних задачах його використовувати нераціонально.
Інший підхід до методу сортування вставками здійснює алгоритм прямих вставок. Головною ознакою його є те, що при виконанні пошуку правильного місця для елемента ми проходимо по впорядкованій частині масиву, а не по невпорядкованій, як у попередніх методах.

На початку роботи ми розбиваємо масив на дві частини: відсортовану та невідсортовану. Оскільки ми не можемо визначити, скільки елементів у масиві вже впорядковано, розбиваємо його таким чином, щоб впорядкована частина містила тільки один (перший) елемент (адже очевидно, що один елемент сам із собою завжди впорядкований). Далі ми починаємо вибирати елементи із невпорядкованої частини по черзі та додавати їх до вже впорядкованої частини, не порушуючи впорядкованості. Для цього при черговому проході ми пересуваємо поточний (i-й) елемент вліво доти, доки частина масиву від 1 до i не стане впорядкованою, тобто поки поточний елемент не займе своє правильне місце у відсортованій частині масиву. Це відбудеться тоді, коли сусідні елементи ліворуч і праворуч поточного стоятимуть “правильно” відносно нього. Таким чином ми “вставляємо” елемент у відсортовану частину масиву, звідки й пішла назва методу.

Реалізація методу наступна:

Procedure InsertSort (var A:IntArray;N:integer);
Var i,j:integer;

Begin
  For i:=2 to N do
  Begin
    j:=i;

    While (A[j]<A[j-1])and(j>1) do
    Begin
      Swap(A[j],A[j-1]);

      Dec(j);

    End;

  End;
End;

Цей метод не є досить швидкодійним, оскільки у ньому, як і в сортуванні “бульбашкою”, виконується багато обмінів сусідніх елементів. Його можна удосконалити, запам’ятовуючи елемент, для якого розшукується місце у вже відсортованій частині, в окремій змінній Etalon. Далі переглядаються всі елементи ліворуч від еталонного і якщо вони більші за еталон, виконується зсув праворуч. Ці дії виконуються, доки не буде знайдено елемент, що не перевищує еталон, або не буде досягнуть ліву границю масиву. Знайдене місце і буде місцем для чергового елементу, що розглядається, тому залишається вставити його на своє місце. Програмна реалізація описаного методу буде наступною:
Procedure InsertSort (var A:IntArray;N:integer);
Var i,j,Etalon:integer;

Begin
  For i:=2 to N do
  Begin
    j:=i;Etalon:=A[i];
    While (Etalon<A[j-1])and(j>1) do
    Begin
      A[j]:=A[j-1];

      Dec(j);

    End;
    A[j]:=Etalon;
  End;
End;

Обидві попередні модифікації алгоритму мали один суттєвий недолік: в них порівнювались сусідні елементи і тому впорядкування на великих наборах даних виконувалось досить повільно.
У 1959 році Д.Шелл запропонував удосконалення сортування за допомогою прямих вставок. Він подав ідею порівнювати елементи, що знаходяться на певній відстані один від одного.
Розглянемо конкретний приклад. Нехай масив має наступний вигляд:

Спочатку окремо згрупуємо і впорядкуємо елементи, які знаходяться на відстані 4 (четверне впорядкування). У поданому прикладі вісім елементів і кожна група складається рівно з двох елементів.

Після першого проходу елементи перегрупуються – тепер в одній групі будуть елементи, які знаходяться на відстані 2 (подвійне впорядкування):

Нарешті на третьому проході виконується звичайне (одинарне) впорядкування.

Для описаного алгоритму зрозуміло, що початковим кроком методу Шелла можна брати h=N/2. Після кожного проходу можна зменшувати крок вдвічі.
На перший погляд виникають сумніви: якщо необхідно виконати кілька процесів впорядкування, причому у кожному з них приймають участь усі елементи масиву, то чи не сповліьнять вони алгоритм замість того, щоб його прискорити? Але зважте: на кожному кроці або впорядковується незначна кількість елементів, або елементи вже достатньо добре впорядковані і не потребують великої кількості перестановок для остаточного впорядкування. Як показує практика, метод прямих вставок дає хороші швидкісні характеристики саме за таких умов.
Procedure ShellSort (var A:IntArray;N:integer);
Var i,j,h:integer;

    Temp:integer;

Begin

 h:=N div 2;

 While h>0 do

 Begin

   For i:=h to N-h+1 do

   Begin

     j:=i; Temp:=A[i];

     While (j>=h)and(Temp<A[j-h]) do

     Begin

       A[j]:=A[j-h];

       Dec (j,h);

     End;

     A[j]:=Temp;

   End;
   h:=h div 2;

 End;

End;

Як видно, ідея методу Шелла полягає в упорядкуванні вхідних даних таким чином, що сукупність h-х елементів масиву є впорядкованим підмасивом. Іншими словами, є h незалежно відсортованих взаємнопроникних частин вхідного масиву. Використання описаного алгоритму для будь-якого ряду значень “кроків” порівняння h, останній з яких дорівнює одиниці, завершується отриманням упорядкованого файлу.

Один з методів його реалізації являє собою незалежне сортування кожної з h-підгруп методом прямих вставок для значень з ряду h більших за одиницю. Але яку послідовність кроків слід обрати?

Навіть у поданому вище прикладі було видно неоптимальність послідовності кроків, запропонованої автором методу. Перш за все це той факт, що елементи масиву, що стоять на парних місцях до останнього проходу не порівнюються з тими, що стоять на непарних місцях.

Було винайдено багато послідовностей кроків h, які добре зарекомендували себе, але найкращого з них не обереш. Доведено лише, що для оптимальності кроки не повинні бути дільниками один одного. В основному використовують спадну геометричну прогресію, оскільки у цьому випадку кількість кроків буде близькою до log N. Практичне покращення ефективності алгоритму при виборі правильної послідовності оцінюється у 20-25%, але сам цей вибір є цікавою задачею.
Так, наприклад, у 1969 році Дональд Е. Кнут запропонував таку послідовність кроків:

1
4
13
40
121
364
1093
3280
9841 ...
Вона одержується досить просто (починаючи з 1 для отримання наступного числа помножуємо дане на 3 і додаємо 1), але забезпечує ефективну роботу у досить великих об’ємах даних (порядку 

 операцій порівняння).

При використанні цієї послідовності  алгоритм прямих вставок модифікується до наступного вигляду:

Procedure ShellSort (var A:IntArray;N:integer);
Var i,j,h:integer;

    Temp:integer;

Begin

 h:=1;

 While h<=N div 9 do h:=3*h+1;

 While h>0 do

 Begin

   For i:=h to N-h+1 do

   Begin

     j:=i; Temp:=A[i];

     While (j>=h)and(Temp<A[j-h]) do

     Begin

       A[j]:=A[j-h];

       Dec (j,h);

     End;

     A[j]:=Temp;

   End;
   h:=h div 3;

 End;

End;

Як видно, цей алгоритм не відрізняється від запропонованого Шеллом, за виключенням операцій по знаходженню кроку h.
Також дійсно оптимальним можна вважати такий ряд:

1
8
23
77
281
1073
4193
16577 ...,

тобто за законом 2(2i + 3(2i + 1. При цьому алгоритм в середньому робитиме порядку 

 операцій порівняння.

Ще оптимальнішою за статистикою є послідовність:

1
5
19
41
109
209
505
929
2161
3905 ...

Вона одержується при об’єднанні послідовностей 9((2 i) 2 –9(2i + 1 та 2(2i –3(2i+1 для i>0.

Існує і ще один вид послідовностей кроків – він базується на тому, що кроки не взаємно прості. У 1971 році Пратт опублікував ідею використання своєрідного трикутника чисел, у якому кожне число у два рази більше за те, що стоїть зверху праворуч, і у три рази більше за те, що стоїть зверху ліворуч. Ось таким чином:
1

2     3

4       6      9

8     12    18    27

16    24    36    54     81

32    48    72   108   162   243

64   96   144   216   324   486   729

...

Якщо використати ці числа знизу вгору і справа наліво у рамках сортування методом Шелла, то жоден елемент не буде рухатись більше, ніж на одну позицію в процесі всього сортування!

Проте ця послідовність не дуже ефективна на практиці (сортування виконує порядку Θ(N((log N)2) порівнянь) через те, що використаних чисел-кроків занадто багато.

Швидкі методи сортування
Ми розглянули багато різновидів базових сортувань, але очевидно, що в найгіршому випадку складність майже всіх їх становить Θ(N2), тобто сортування великих об’ємів даних за допомогою поданих вище алгоритмів буде занадто довгим. Що ж робити в таких випадках? Для цього й були винайдені різноманітні методи сортування зі складністю, меншою за Θ(N2) (порядку Θ(N(log N), і навіть порядку Θ(N)). Такі методи впорядковують великі структури даних (до 30000 елементів) менш ніж за секунду. Але майже в кожного з них існують недоліки: від довгого часу написання програми до великої складності у розумінні суті алгоритму, від обмежень на вхідні дані до величезної різниці у часі виконання в кращому й гіршому випадках.

До методів швидкого сортування відносяться наступні:

1) швидке сортування;

2) пірамідальне сортування;
3) сортування “злиттям”;
4) порозрядне сортування;
5) метод розподільних підрахунків.
Розглянемо кожен з них окремо.

Швидке сортування
Базовий алгоритм цього методу сортування був вперше винайдений у 1960 році К.А.Р.Хоаром (C.A.R.Hoare). Він став одним із найпопулярніших перш за все тому, що порівняно нескладний у реалізації, добре працює на різноманітних видах вхідних даних і не вимагає великих системних ресурсів (цей вид відноситься до сортуваннь “на місці”), окрім невеликого додаткового стека. Проте метод, звичайно, має і недоліки: так, він є нестійким (не зберігає взаємний порядок рівних елементів), сильно змінює час роботи в залежності від первинної ступені впорядкованості даних (в кращому випадку лише Θ(2Nlog N) операцій, а в гіршому аж N2/2), а найменша помилка у реалізації може призвести до абсолютно неправильної роботи програми.

Перейдемо до базового алгоритму швидкого сортування. Він функціонує за стародавнім принципом “розділяй і володарюй”. Суть його полягає в тому, що серед елементів масиву один обирається за еталон (традиційно, середній, хоча не обов’язково: як побачимо далі, за еталон з метою оптимізації  різні модифікації даного методу беруть різні елементи), відносно якого масив розбивається на дві взаємнонезалежні частини, тобто з однієї сторони (ліворуч) розміщуються елементи, що не перевищують еталон, з іншої (праворуч) – більші за еталон. Таким чином, якщо уважно придивитися, як мінімум один (еталонний) елемент займає своє кінцеве правильне місце у масиві, після чого він вже не розглядається і не рухається з місця. Після розбиття процес повторюється для обох отриманих частин окремо. Так відбувається, доки усі елементи масиву не займуть правильних положень (масив стане остаточно впорядкованим).

Оскільки один той самий процес повторюється для підмасивів різної довжини, при програмуванні цього алгоритму зручно використовувати рекурсивне оформлення підпрограми. Загалом очевидно, що алгоритм має бути наступним:

Procedure QuickSort (var A:IntArray;L,R:integer);

Var i,j:integer;

Begin
  <Вибір еталонного елемента і  >
  <розбиття масиву відносно нього>

  QuickSort (A,L,j);

  QuickSort (A,i,R);
End;

Розбиття масиву виконується з використанням наступної стратегії. Перш за все еталоном обирається певний елемент з діапазону L..R (як правило, це центральний за розміщенням елемент). Далі здійснюється рух по масиву від лівої границі (L) у напрямку до правої границі (R) до тих пір, поки не знайдеться елемент не зі своєї “половини”, тобто елемент, що перевищує за значенням еталон. Одночасно виконується рух від правої границі (R) до лівої границі (L) доки теж не знайдеться елемент не з своєї “половини”, тобто елемент, що менший за еталон. Ці елементи, очевидно, стоять не на своїх місцях, тому вони обмінюються місцями. Продовжуємо рух по масиву, поки не переконаємося, що зліва від лівого “курсора” і (тобто курсора, що „керує” рухом зліва направо) і справа від правого j (курсора, що „керує” рухом справа наліво) немає жодного елемента “не зі своєї половини”. В цей момент курсори перетнуться і в результаті частини (L..j) та (i..R) стануть взаємнонезалежними, а еталон відповідно стане на кінцеве своє положення.
Тепер процес необхідно повторити для кожної зі вказаних частин, а для цього необхідно з середини процедури викликати її саму, тільки з іншими вхідними параметрами. Це і є рекурсія. Нагадаємо, що рекурсія – це виклик підпрограмою самої себе і тому, щоб процес не став нескінченним і не виникло явище переповнення стеку, необхідно обов’язково визначити момент, коли почнеться вихід (повернення) з рекурсії. У нашому випадку очевидно, що вихід з рекурсії повинен початися в той момент, коли чергові утворені підмасиви будуть мати довжину в один елемент, тобто коли ліва границя підмасиву не менша за праву границю. Остаточний алгоритм може виглядати ось так:

Procedure QuickSort (var A:IntArray;L,R:integer);

Var i,j,Etalon:integer;
Begin
 If L>=R then Exit;

 Etalon:=A[(L+R) div 2];

 i:=L; j:=R;

 Repeat

  While A[i]<Etalon do Inc (i);

  While A[j]>Etalon do Dec (j);

  If i<j 
  then Begin

        Swap(A[i],A[j]);

        Inc(i); Dec(j);

       End;

 Until i>=j;

 QuickSort (A,L,j);

 QuickSort (A,i,R);

End;

Існує багато модифікацій методу, але ми розглянемо три основні, які суттєво підвищують ефективність швидкого сортування.

Перш за все ви, мабуть, вже задавали собі питання: “Чому за еталон беруть середній за розташуванням елемент A[(L+R)/2]?” Дійсно, ми використали саме його лише тому, що в цьому випадку не потрібно накладати на курсори додаткові умови для запобігання виходу за межі підмасиву L..R у деяких випадках, але хто сказав, що цей вибір оптимальний? Адже може статися, що ми оберемо за еталон максимум або мінімум, тоді проходи будуть марними, а розміри невідсортованої “незалежної” частини зменшаться лише на один елемент. Як же бути? Звичайно, вибрати випадково “золоту середину” (медіану) з усіх елементів ми не можемо, але хоча б обрати з трьох – для визначеності, першого (L), останнього (R) та середнього ((L+R)/2), – можна?

А для того, щоб серед них обрати середнє значення, відсортуємо їх трьома командами розгалуження:

... ... ... ...

If A[L]>A[(L+R) div 2] 
then Swap (A[L],A[(L+R) div 2]);

If A[(L+R) div 2]>A[R] 
then Swap (A[(L+R) div 2],A[R]);

If A[L]>A[(L+R) div 2] 
then Swap (A[L],A[(L+R) div 2]);

Etalon:=A[(L+R) div 2];

... ... ... ...

Якщо ви тестували попередній метод на різних типах великих масивів, то могли побачити очевидну зайву річ: навіщо розглядати в подальшому елементи, значення яких дорівнює еталону, якщо можна одразу поставити їх поряд на свої місця, визначивши правильні границі “меншої” та “більшої” сортованої підгрупи?

Таке розбиття на три групи (менші, рівні та більші за еталон елементи) важче за те, яким ми користувалися раніше, але теж доволі наглядне і легкореалізовне. Оригінальний метод був запропонований в 1993 році Бентлі (Bentley) та МакІлроєм (McIlroy). Це модифікація стандартного розбиття, яка заключається в наступному: рівні еталону елементи накопичуються у кінцях масиву, а коли курсори перетнуться і точне місцезнаходження цих елементів стане відомим, усі вони переміщуються у цю позицію. Цей метод незначно вповільнить розбиття, але загалом сортування виконуватиметься швидше, ніж стандартне.

У розділяючому циклі, коли курсори (i, j) знайшли два елементи “не зі своїх половин” і обміняли ці елементи, виконується перевірка кожного з них на рівність еталону і, якщо рівність має місце, то елемент переміщується в p-ту (q-ту) позицію від краю масива (щоб елементи не накладалися).

Procedure QuickSort (var A:IntArray;L,R:integer);
Var Etalon,i,j,k,p,q:integer;

Begin
 If L>=R then Exit;

 Etalon:=A[(L+R) div 2];

 i:=L; j:=R;

 p:=L-1; q:=R+1;

 Repeat

  While A[i]<Etalon do Inc(i);

  While A[j]>Etalon do Dec(j);

  If i<j 
  then Begin

        Swap(A[i],A[j]);

        Inc(i); Dec(j);

        If A[i]=Etalon 
        then Begin

              Inc(p);

              Swap(A[p],A[i]);

             End;

        If A[j]=Etalon 
        then Begin

              Dec(q);

              Swap(A[q],A[j]);

             End;

       End;

 Until i>=j;

 For k:=L to p do

 Begin

   Swap(A[k],A[j]);

   Dec(j);

 End;

 For k:=R-L downto q do

 Begin

   Swap(A[k],A[i]);

   Inc(i);

 End;

 QuickSort(A,L,j);

 QuickSort(A,i,R);

End;
Ще одне помітне підвищення ефективності сортування випливає з того факту, що наша рекурсивна програма гарантовано викликає саму себе для обробки багатьох невеличких частин вхідних даних. Звідси логічно випливає, що для впорядкування таких частинок бажано використовувати найефективніший метод сортування, яким, нажаль, саме швидке сортування не є.

Тому, хто розв’язував колись олімпіадні задачі, знайоме таке поняття, як комбінування алгоритмів. Тож і тепер ми в алгоритмі швидкого сортування використаємо інший метод для сортування малих частин масиву – методом вставки. Перевірку на метод сортування слід поставити замість перевірки на коректність границь сортування:

... ... ... ...

If R-L<=MinLen 
then Begin

      InsertSort(A,L,R);

      Exit;

     End;

Etalon:=A[(L+R) div 2];

... ... ... ...

Тут константа MinLen може бути визначена емпіричним (експериментальним) шляхом чи аналізом. Це значення було визначене оптимальним, в залежності від кількості елементів у масиві, в інтервалі від 5 до 20 (точне значення не принципове). Зверніть також увагу на те, що у порівнянні з наведеним вище алгоритмом прямої вставки, запропонований виклик InsertSort(A,L,R); містить три параметри, що передаються, а не два. Справа в тому, що наведений раніше алгоритм впорядковував всі елементи масиву, починаючи з першого, а у комбінованому алгоритмі сортування прямими вставками необхідно застосовувати для підмасиву з лівою та правою границями. Сподіваємось, що внести незначні зміни у підпрограму InsertSort Ви зможете самостійно.
Дещо легший і трохи швидший порівняно з попереднім спосіб полягає в тому, що частинки менші за MinLen просто ігноруються (if R-L<=MinLen then Exit;), а після завершення неповного сортування весь масив даних впорядковується найкращим в даному разі методом – прямими вставками.

Пірамідальне сортування
Пірамідальне сортуваня (або сортування за допомогою купи) використовує таку структуру даних, як врівноважене бінарне дерево або, як його ще називають, купа.

Купою називають масив, що має деякі властивості. Щоб легше було їх уявити, представимо масив, як бінарне дерево (дивись рис. 1). Кожна його вершина відповідає елементу масиву. У вершини з номером i дітьми, якщо вони існують, будуть елементи 2i та 2i+1, а батьком – i div 2. Перший елемент масиву буде коренем дерева.

Головною властивістю купи, яка має обов’язково виконуватись, є правило: A[i div 2](A[i] (окрім кореня). Звідси випливає, що найбільший елемент дерева знаходиться у корені дерева.

Висотою вершини дерева називається кількість ребер у найдовшому шляху з цієї вершини вниз до листа (вершини, у якої немає синів). Висота дерева, що складає купу з N елементів, дорівнює приблизно log2 N.

Суть методу впорядкування за допомогою купи базується на тому факті, що корінь правильно побудованої купи буде максимальним її елементом. Тому перед впорядкуванням ми робимо з масиву правильну купу, а далі, у процесі сортування, помістивши кореневий (перший) елемент на своє місце, відновлюємо серед решти елементів масиву основну властивість купи.

Алгоритм здається зовсім простим:

Procedure HeapSort (var A:IntArray;N:integer);
Begin

 BuildHeap (A,N);

 For i:=N downto 2 do

 Begin

  Swap (A[1],A[i]);

  RestoreHeap (A,i-1,N);

 End;

End;
При первинному створенні купи з масиву даних (процедура BuildHeap) ми проходимо по всім елементам-батькам масиву і відновлюємо у їхніх гілках основну властивість купи:

Procedure BuildHeap (var A:IntArray;N:integer);

Var i:integer;

Begin

  for i:=N div 2 downto 1 do

     RestoreHeap (i,N);

End;
Коли ж ми прибрали кореневий елемент, виникає необхідність відновити правильну купу (підпрограма RestoreHeap). Ідея цього відновлення проста: якщо основна властивість у новому кореневому елементі не виконується, то більший з синів обмінюється з батьком, після чого основна властивість перевіряється і у сина.

Procedure RestoreHeap (var A:IntArray;C,N:integer);
Var MaxSon:integer;

Begin

 MaxSon:=C;

 If (C*2<=N)and(A[C]<A[C*2])

 then MaxSon:=C*2;

 If (C*2+1<=N)and(A[MaxSon]<A[C*2+1])

 then MaxSon:=C*2+1;

 If MaxSon<>C 
 then Begin

       Swap (A[C],A[MaxSon]);

       RestoreHeap (A,MaxSon,N);

      End;

End;
Сортування “злиттям”
Одною з найпривабливіших властивостей цього методу сортування є те, що він впорядковує N елементів за час, пропорційний Θ(N log N), незалежно від характеру вхідних даних. Метод є стійким і це схиляє чашку терезів у його сторону в тих випадках, коли дана обставина є дуже важливою. Але одним із найголовніших недоліків його є той факт, що в простій реалізації необхідний додатковий масив для злиття. Цю перешкоду можна і обійти, але зробити це досить важко, причому знадобляться додаткові витрати часу, які в загальному випадку себе не виправдовують. Один з варіантів обходу додаткових витрат був запропонований Бетчером, але його ми розглянемо пізніше.

В основі базового алгоритму сортування лежить така корисна і необхідна у деяких задачах операція, як злиття двох впорядкованих підмасивів у один (теж впорядкований). Це обернена до використаної у швидкому сортуванні (коли ми розбивали один масив на дві частини) операція.

Ця операція реалізується досить легко і зрозуміло. Можна уявити два підмасиви у вигляді стеків (створити курсори, що вказуватимуть номери “голів” і збільшувати їх, коли з стека “забирають” елемент). Тоді достатньо у черговий елемент нового масиву заносити менше значення з двох “голів” (або, якщо один зі стеків став порожнім, значення “голови” іншого стека).

Під час сортування ми “зливатимемо” дві сусідні частини одного масиву вхідних даних, тому в підпрограму Merge передаємо лише границі підмасивів (L – початок першої частини; M – початок другої частини; R – кінець другої частини):

Procedure Merge (var A:IntArray;L,M,R:integer);
Var i,Cur1,Cur2:integer;

    aux:array [1..MaxN] of integer;

Begin

  Cur1:=L; Cur2:=M+1;

  For i:=L to R do
      Begin

        If Cur1>M-1 then

           Begin

              aux[i]:=A[Cur2];

              Inc (Cur2);

           End else

        If Cur2>R then

           Begin

              aux[i]:=A[Cur1];

              Inc (Cur1);

           End else

        If A[Cur1]>A[Cur2] Then

           Begin

             aux[i]:=A[Cur1];

             Inc (Cur1);

           End else

           Begin

             aux[i]:=A[Cur2];

             Inc (Cur2);

           End;

      End;

  For i:=L to R do A[i]:=aux[i];

End;
Самого ж сортування “злиттям” розрізняють декілька методів.

Злиття зверху донизу. Реалізація методу, як і попередні швидкі сортування, містить рекурсію. Вона полягає у тому, що спочатку впорядковуються дві частини масиву (L..(L+R)/2 та (L+R)/2+1..R), а потім ці частини зливаються за допомогою описаного вище алгоритму.

Програма прозора:

Procedure MergeSort (var A:IntArray;L,R:integer);
Begin

  If L>=R then Exit;

  MergeSort (A,L,(L+R) div 2);

  MergeSort (A,(L+R) div 2+1,R);

  Merge (A,L,(L+R) div 2+1,R);

End;
Злиття знизу догори. Цей метод не використовує рекурсію. Спочатку весь масив розбивається (умовно) на частини довжиною в один елемент, які після цього попарно зливаються. Коли маленькі частинки злито, процес повторюється для отриманих вдвічі більших частин.

Procedure MergeSort (var A:IntArray;N:integer);
Var m,i:integer;

Begin

  m:=1;

  While m<N do

        Begin

          i:=0;

          While i+m<=N do

            Begin

              Merge (A,i+1,i+m,Min (i+2*m,N));

              i:=i+2*m;

            End;

          m:=2*m;

        End;

  Merge (A,1,N div 2,N);

End;

Парно-непарне злиття. Алгоримт був розроблений Бетчером у 1986 році під назвою Batcher’s odd-even sort.

Цей метод є неадаптивним (послідовність виконуваних дій залежить тільки від кількості вхідних даних, а не від значень ключів), тобто він може бути записаний у вигляді простого переліку операцій порівняння-обміну (як, наприклад, сортування “бульбашки” чи Праттова версія сортування методом Шелла).

В основі його лежать дві обернені операції: ідеальне тасування та зворотнє ідеальне тасування. Це метод спеціального призначення і його рідко використовують, перш за все тому, що зрозуміти, чому він працює правильно, досить важко.

Операція ідеального тасування переупорядковує масив так, як може перетасувати колоду карт тільки великий спеціаліст цієї справи: колода ділиться точно пополам, після чого карти по одній (починаючи з верхньої половини колоди) беруться з кожної частини і складаються в одну стопку.

Procedure Shuffle (var A:IntArray;L,R:integer);
Var i:integer;

    aux:array [1..MaxN] of integer;

Begin
  for i:=0 to R-L do
    if i mod 2 = 0

     then aux[l+i]:=A[l+i div 2]
     else aux[l+i]:=A[(l+r) div 2+1+i div 2];
  for i:=L to R do A[i]:=aux[i];

End;
Зворотнє ідеальне тасування виконує обернену операцію: з колоди карти беруться по порядку по одній і по черзі кладуться то поверх нової стопки, то під низ її.

Procedure Unshuffle (var A:IntArray;L,R:integer);
Var i:integer;

    aux:array [1..MaxN] of integer;

Begin
  For i:=0 to (L+R) div 2-L do
      Begin
        aux[L+i]:=A[L+i*2];

        aux[(L+R) div 2+1+i]:=A[L+1+i*2];
      End;

  For i:=L to R do A[i]:=aux[i];

End;

Саме ж злиття Бетчера виконується за такою схемою:

1) зворотне ідеальне тасування (розбиття групи L..R на незалежні підгрупи вдвічі меншого розміру);

2) виконання того ж злиття Бетчера, але для кожної підгрупи елементів L..(L+R)/2 та (L+R)/2+1..R окремо;

3) ідеальне тасування (повернення елементів підгруп на початкове місце);

4) перевірка правильності розташування елементів у отриманому масиві та виправлення "помилок" сортування.

Procedure Merge (var A:IntArray;L,R:integer);
Var i:integer;

Begin

  if (R=L+1)and(A[L]>A[R]) then

     Swap (A[L],A[R]);

  if R<=L+1 then Exit;

  Unshuffle (A,L,R);

  Merge (A,L,(L+R) div 2);

  Merge (A,(L+R) div 2+1,R);

  Shuffle (A,L,R);

  i:=L+1;

  while i<R do
        Begin
          If A[i]>A[i+1] then
             Swap (A[i],A[i+1]);
          Inc (i,2);

        End;

End;
А алгоритм сортування Бетчера відрізняється від сортування злиттям тільки заміною звичайної процедури злиття запропонованою вище.
Іноді виникає ситуація, коли елементи масиву даних мають дещо незвичайні властивості, які можна використати для одержання надшвидких методів сортування аж до лінійної складності (Θ(N)).
Порозрядне сортування
У багатьох випадках ключі сортування, які використовуються для визначення взаємного порядку елементів, мають досить складну природу. Так, наприклад, при впорядкуванні за алфавітом прізвищ у базі даних учнів, для порівняння двох елементів необхідно пройти по порядку по їхнім буквам, аж поки не знайдеться пара нерівних букв, за якими можна буде визначити порядок учнів у списку. Звичайно, можна робити таке порівняння під час виконання порівняння, але це вповільнить процес сортування у декілька разів.
Колись, коли єдиними зберігачами інформації (баз даних, програм) були перфокарти інженери користувалися наступним алгоритмом для їх впорядкування.

Сортуюча машина могла розподілити перфокарти на групи за розрядом, який вкаже інженер. При цьому в першій групі будуть знаходитися ті перфокарти, вказаний розряд яких дорівнює 0, в другій, відповідно, ті, вказаний розряд яких дорівнює 1 і т.д.

Очевидно, що для правильного остаточного впорядкування перфокарт необхідно зробити наступне. Усі перфокарти помістити в сортуючу машину і відсортувати їх за першим розрядом. Кожну з отриманих груп по черзі помістити до сортуючої машини і відсортувати за другим розрядом. Далі повторюємо ті ж дії для всіх наступних розрядів. Отриману велику кількість груп по порядку складаємо в одну – вона стане відсортованою.

Методи сортування, побудовані на обробці чисел по одній порції за раз, називаються порозрядними (radix), а описаний вище метод – MSD radix sort (most significant digit radix sort – порозрядне сортування за найбільшим розрядом).

Вважатимемо, що ми сортуємо масив строк за D-ю буквою.  В процесі впорядкування викликатиметься ця ж процедура, але для меншого підмасиву і для наступної букви. Реалізація використовує дещо змінений метод підрахунків:

Procedure MSDSort (var A:IntArray;L,R,D:integer);
Var aux:IntArray;

    c:char;

    i,count,oldcount:integer;

Begin
  сount:=L-1;

  For c:=#0 to #255 do
      Begin

        oldcount:=count+1;

        For i:=L to R do
           If Length (A[i])>=D then
            If A[i,D]=c then
               Begin
                 Inc (count);

                 aux[count]:=A[i];

               End;

        if count>oldcount then

           DigSort (aux,oldcount,count,D+1);

      End;

  For i:=L to R do A[i]:=aux[i];

End;

Цю підпрограму основне тіло програми викликає для усього масиву (L=1, R=N) і за першим розрядом (D=1). Процедура в свою чергу рекурсивно викликає сама себе для інших розрядів, для менших під масивів вхідного масиву, якщо це необхідно.

Інший метод порозрядного сортування називається LSD radix sort (least significant digit radix sort – порозрядне сортування за найменшим розрядом). Він не такий природній та інтуїтивно зрозумілий для декого, але теж не складний.

Тут, навідміну від попереднього методу, масив не розбивається на частини, а сортування починається з молодшого розряду. Відповідно, у випадку, коли слова мають неоднакову довжину, виникає необхідність доповнити їх проміжками (або символом #0 чи будь-яким іншим, але одним і тим же і, бажано, невидимим, щоб не було проблем у виведенні масиву на екран).

Метод, за допомогою якого дані впорядковуються за одним розрядом, має бути стійким, інакше сортування працювати не буде. Таким методом є, наприклад, метод розподільних підрахунків:

Procedure LSDSort (var A:IntArray;N,D:integer);
var aux:IntArray;

    c:char;

    i,count:integer;

Begin
  if D<=0 then Exit;

  count:=0;

  For c:=#0 to #255 do
      Begin
        For i:=1 to N do
            If A[i,D]=c then
               Begin
                 Inc (count);

                 aux[count]:=A[i];

               End;

      End;

  For i:=1 to N do A[i]:=aux[i];

End;

В основній програмі, як було сказано вище, спочатку всі слова вхідного масиву робляться однієї довжини, а потім у циклі, від останнього до першого розряду, викликається підпрограма LSDSort.
Метод розподільних підрахунків

Підвищення ефективності деяких алгоритмів сортування досягається за рахунок використання специфічних властивостей елементів.

Наприклад, сформулюємо задачу так: впорядкувати масив з N елементів, що приймають  цілі значення 0..M. Метод розподільних підрахунків дозволяє зробити це досить легко й швидко, але при допомозі додаткового масиву. Ідея його заключається у наступних діях:
1) підрахувати кількість елементів з кожним конкретним значенням 0..M;
2) визначити кількість елементів масиву, менших або рівних кожному значенню;
3) використати лічильники як індекси при розподілі значень і запам’ятовуванні їх у масиві B;
4) перезаписати масив A значеннями з масиву B.

В процедуру сортування передаємо разом з масивом і його розмірами ще й максимальне значення (M), яке можуть приймати його елементи:

Procedure Counting (var A:IntArray;N,M:integer);
Var i:integer;

    Count,B:IntArray;

Begin

  For i:=0 to M do Count[i]:=0;

  For i:=1 to N do Inc (Count[A[i]+1]);

  For i:=1 to M do

      Count[i]:=Count[i]+Count[i-1];

  For i:=1 to N do

      Begin
        B[Count[A[i]]]:=A[i];

        Inc (Count[A[i]]);

      End;

  For i:=1 to N do A[i]:=B[i-1];

End;
Рис. 1. Бінарне дерево

